
HP-35s Calculator Program Closure 3

Three-Point Horizontal Resection Reduction Program

Programmer: Dr. Bill Hazelton

Date: October, 2007. Version:  1.0

Line Instruction Display User Programming Instructions
R001 LBL R
R002 CLSTK
R003 FS?  10
R004 GTO  R008
R005 SF  1
R006 SF 10
R007 GTO  R009
R008 CF  1
R009 RESECTION
R010 PSE
R011 ENTER LEFT X (Key in using EQN  RCL E  RCL N etc.)
R012 PSE
R013 INPUT  X
R014 STO  A
R015 ENTER LEFT Y (Key in using EQN  RCL E  RCL N etc.)
R016 PSE
R017 INPUT  Y
R018 STO  B
R019 ENTER MID X (Key in using EQN  RCL E  RCL N etc.)
R020 PSE
R021 INPUT  X
R022 STO  C
R023 ENTER MID Y (Key in using EQN  RCL E  RCL N etc.)
R024 PSE
R025 INPUT  Y
R026 STO  D
R027 ENTER RIGHT X (Key in using EQN  RCL E  RCL N etc.)
R028 PSE
R029 INPUT  X
R030 STO  E
R031 ENTER RIGHT Y (Key in using EQN  RCL E  RCL N etc.)
R032 PSE
R033 INPUT  Y
R034 STO  F
R035 ENTER ALPHA (Key in using EQN  RCL E  RCL N etc.)
R036 PSE
R037 INPUT  X
R038 HMS→
R039 STO  G

R040 ENTER BETA  (Key in using EQN  RCL E  RCL N etc.)
R041 PSE
R042 INPUT  X
R043 HMS→
R044 STO  H
R045 RCL  B
R046 RCL–  D
R047 RCL  A
R048 RCL–  C
R049 0  i  1 Press the zero key, then i, then 1.
R050 ×
R051 +
R052 STO  L
R053 RCL  F
R054 RCL–  D
R055 RCL  E
R056 RCL–  C
R057 0  i  1 Press the zero key, then i, then 1.
R058 ×
R059 +
R060 STO  K
R061 360
R062 STO  Z
R063 RCL  L
R064 ARG
R065 RCL  K
R066 ARG
R067 –
R068 x < 0 ?
R069 RCL+  Z
R070 STO  I
R071 RCL+  G
R072 RCL+  H
R073 RCL  Z
R074 x < > y
R075 –
R076 STO  S
R077 RCL  L
R078 ABS
R079 RCL  H
R080 SIN
R081 ×
R082 RCL  K
R083 ABS
R084 ÷
R085 RCL  G
R086 SIN

R087 ÷
R088 RCL  S
R089 SIN
R090 ÷
R091 RCL  S
R092 TAN
R093 1/x
R094 +
R095 1/x
R096 ATAN
R097 STO  X
R098 RCL  L
R099 ARG
R100 180
R101 +
R102 RCL+  G
R103 RCL+  X
R104 STO  Y
R105 RCL  L
R106 ABS
R107 RCL  X
R108 SIN
R109 ×
R110 RCL  G
R111 SIN
R112 ÷
R113 STO  J
R114 RCL  Y
R115 SIN
R116 ×
R117 STO  P
R118 RCL  J
R119 RCL  Y
R120 COS
R121 ×
R122 STO  Q
R123 RCL  P
R124 RCL+  C
R125 STO  X
R126 UNKNOWN X = (Key in using EQN  RCL U  RCL N etc.)
R127 PSE
R128 VIEW  X
R129 UNKNOWN Y = (Key in using EQN  RCL U  RCL N etc.)
R130 PSE
R131 RCL  Q
R132 RCL+  D
R133 STO  Y

R134 VIEW Y
R135 RCL  I
R136 RCL+  G
R137 RCL+  H
R138 →HMS
R139 CHECK VALUE = (Key in using EQN  RCL C  RCL H etc.)
R140 PSE
R141 STOP
R142 FS?  1
R143 CF 10
R144 RTN

Notes

(1) Horizontal 3-point resection solution, based on measuring two angles between
three known points at an unknown point, the location of which is to be computed.

(2) Brief prompts are provided before each requirement for data entry, as well as
before results are displayed. Each prompt shows for about 1 second, and is then
replaced by the value or request for input.

(3) Co-ordinates of the unknown point are displayed following brief prompts. They
are also stored in storage registers X and Y for later retrieval.

(4) Angles are entered and displayed in HP notation, i.e., DDD.MMSS. Internal
storage of angles and azimuths is in decimal degrees. Internal storage of lines
uses the calculator’s complex number format.

Theory

This 2-D resection uses Ormsby’s solution. In the discussion below, A is the left point, B is the
middle point, C is the right point, and P is the unknown point. The left angle is alpha (α) and the
right angle is beta (β). The interior angle at B is gamma (γ). The angle at point A is x, which is
the first objective of the solution. A diagram is shown on the next page.

α and β are angles observed from the point P to points A, B and C, whose co-ordinate are known.

BP  =  
AB sin x

sin α     =  
BC sin y

sin β   

and (x + y)  =  (360° – (α + β + γ))  =  s

AB
sin α   sin x  =  

BC
sin β   sin (s – x)  =  

BC
sin β (sin s cos x – cos s sin x)  

AB
sin α   sin x  =  

BC
sin β   sin s cos x – 

BC
sin β   cos s sin x

sin x (
AB

sin α    +  
BC

sin β   cos s)  =  
BC

sin β   sin s cos x

( AB
sin α    +  

BC
sin β   cos s) 

sin β
BC sin s    =  cot x

AB sinβ
BC sin α sin s    +  

BC cos s sin β
BC sin s sin β     =  cot x

AB sin β
BC sin α sin s    +  cot s  =  cot x [this is the equation solved first]

y  =  s  –  x

With x and y determined, the sides AP, BP and CP can be calculated and hence the co-ordinates
of P, as follows:

The azimuth of BP (AzBP) can be determined using   AzBP  =  AzAB  +  α  +  x

The length of BP can be determined using     BP  =  
AB sin x

sin α   

A

B

C

P

x y

α β

γ

Knowing the co-ordinates of B, AzBP and BP, the co-ordinates of P can be easily computed. As a
check, the equivalent solution can be obtain through the sides AP or CP, or using the angle y.
Note that if P is close the danger circle, a solution will still be obtained, but the sum of α + β + γ
will be close to 180°, probably in the range 175° to 185°. In this case, the solution will be highly
sensitive to changes in α and β. If the solution is close to the danger circle, recomputed with the
angles changed by about their precision and see how much the resulting co-ordinates change. It
can be quite surprising! To facilitate this, press GTO R035, then R/S, to run the program with the
same known points, but you can enter different observed angles.

Azimuths in HP notation are used. Arbitrary co-ordinates are satisfactory. Plane surveying
assumptions apply. The program uses no error checking on entered data. A check is made by
showing the sum  α + β + γ. If this is close to 180°, the unknown point lies close to the danger
circle and the result is highly suspect.

Sample Computation 1

Known Points

Point Name X Y

Point A –25.336 778.136

Point B –27.465 1179.927

Point C –30.297 1555.643

Angles Left (α)  =  136° 35' 26"

Right (β)  =  27° 19' 24"

Results Unknown Point (P)  X Co-ordinate  =      26.009

Unknown Point (P)  Y Co-ordinate  =  1101.818

Check Angle  =  344° 02' 32"

Sample Computation 2

Known Points

Point Name X Y

Point A 133.639 1548.712

Point B 158.065 1492.276

Point C 150.267 1353.056

Angles Left (α)  =  5° 01' 48"

Right (β)  =  3° 41' 29"

Results Unknown Point (P)  X Co-ordinate  =     116.784

Unknown Point (P)  Y Co-ordinate  =  1,186.818

Check Angle  =  162° 06' 44"

This is not the ideal arrangement for a resection, as the measured angles are quite small. But the
program will still produce an acceptable result.

This example is provided because the other example has negative co-ordinates and this tends to
increase the chances of incorrect data entry. It happened to me, twice!

Running the Program

Press  XEQ  R  ENTER

Calculator displays RESECTION briefly, so that you know you are running the correct program.

Prompt  ENTER LEFT X briefly, then X?

Enter X Co-ordinate for left known point.

Press R/S.

Prompt  ENTER LEFT Y briefly, then Y?

Enter Y Co-ordinate for left known point.

Press R/S.

Prompt  ENTER MID X briefly, then X?

Enter X Co-ordinate for middle known point.

Press R/S.

Prompt  ENTER MID Y briefly, then Y?

Enter Y Co-ordinate for middle known point.

Press R/S.

Prompt  ENTER RIGHT X briefly, then X?

Enter X Co-ordinate for right known point.

Press R/S.

Prompt  ENTER RIGHT Y briefly, then Y?

Enter Y Co-ordinate for right known point.

Press R/S.

**  Prompt  ENTER ALPHA briefly, then X?

Enter left angle (α) in HP notation.

Press R/S.

Prompt  ENTER BETA briefly, then X?

Enter right angle (β) in HP notation.

Press R/S.

Calculator displays RUNNING while doing the calculations.

Prompt  UNKNOWN X briefly, then X=

X co-ordinate of unknown point (P) is displayed.

Press R/S.

Prompt  UNKNOWN Y briefly, then Y=

Y co-ordinate of unknown point (P) is displayed.

Press R/S.

Prompt  CHECK VALUE briefly.

Sum α + β + γ is displayed in lower line of display in HP notation.

Check that value is not too close to 180°. At least 5 ° away, preferably 15° or more away.

Press R/S to clear flags. Program ends.

If you want to re-run the program with the same fixed points but different angles, press GTO
R035, then R/S, and the program will start from the step labeled ** above, prompting with
ENTER ALPHA. Changing the angles by small amounts can give you a good idea of the
reliability of the solution.

Storage Registers Used

A Left known point – X co-ordinate

B Left known point – Y co-ordinate
C Middle known point – X co-ordinate
D Middle known point – Y co-ordinate
E Right known point – X co-ordinate
F Right known point – Y co-ordinate
G Left measured angle — alpha (α)
H Right measured angle — beta (β)
I Interior angle at Middle known point — gamma (γ)

J Distance from middle point to the unknown point

K Vector from middle to right point (complex number format)
L Vector from middle to left point (complex number format)
P X co-ordinate of unknown point
Q Y co-ordinate of unknown point
S s = x + y in decimal degrees
X Initial inputs, then angle x, then X co-ordinate of unknown point
Y Initial inputs, then azimuth from middle to unknown point, then Y co-ordinate of

unknown point
Z 360

Labels Used

Label R Length  =  581 Checksum  =  17B0

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly.
Use the sample computation to check proper operation after entry.

The program sets flag 10, too allow equations to be displayed as prompts, and at the end of the
program, resets flag 10 to its previous setting. The program uses flag 1 to record the state of flag
10 before the program started.


